

Les impressions

1.Principe de l'édition

L'édition consiste à transformer en mémoire centrale une valeur numérique ou alphanumérique pour en améliorer la présentation en vue de sa visualisation sur un support externe (écran, listing).

Pour cela, deux techniques peuvent être utilisées :

- ◆ **l'insertion** : on insère dans la suite des chiffres des caractères tels que le point, le signe (+ ou -), le symbole monétaire, etc...
- ◆ la **suppression** et le **remplacement** de **zéros** non significatifs.

Le recours à ces moyens s'exprime par des symboles spéciaux qui figurent dans la description de l'image de la rubrique dite **éditée** ou d'édition (clause `PICTURE`).

Pour réaliser l'édition d'une valeur, il suffit alors de la transférer dans une rubrique d'édition ; ceci peut se faire par une instruction `MOVE` (voir règles chap. 3 paragraphe 4.8) ; si la valeur est le résultat d'un calcul, la rubrique d'édition peut être précisée par l'option `GIVING` de la dernière instruction arithmétique.

Attention !

- Une rubrique d'édition est toujours décrite en `USAGE DISPLAY`.

La valeur obtenue après transformation n'est en général plus de type numérique pur : de ce fait elle ne peut intervenir dans des calculs ultérieurs.

2. Les symboles d'insertion

2.1. Le point décimal

Le point `.` remplace le symbole `V` ; il assure l'alignement sur la marque décimale mais en plus génère un caractère `"."`.

Exemple : `E1 PIC 99.99` décrit une rubrique de 5 caractères

```
MOVE 2.51 TO E1 donne 02.51
```

2.2. Les symboles d'insertion simple

Ce sont la virgule (**,**), l'espace (**B**), le zéro (**0**) et le slash (**/**).

Chaque symbole provoque la génération du caractère correspondant à condition de ne pas figurer dans une partie de suppression de zéros non significatifs (cf. § 3 de ce chapitre).

Exemple : E2 PIC 9B999.99 décrit une rubrique de 8 caractères
 MOVE 1500 TO E2 donne 1**b**500.00

Attention !

Ne pas confondre les symboles **,** et **.** ; la virgule n'assure **pas** l'alignement sur la marque décimale.

2.3. Les symboles d'insertion fixe

Ce sont le symbole monétaire (**\$**) et les caractères d'édition du signe (**+**, **-**, **CR**, **DB**). Ils ne doivent apparaître qu'une seule fois par format (PICTURE).

- Le symbole \$ doit être le **premier** symbole du format ; il peut toutefois être précédé d'un symbole + ou d'un symbole -.
- Les symboles + et - se placent en première ou en dernière position du format ; le symbole + provoque l'édition du signe sous forme d'un caractère + (valeur positive) et d'un caractère - (valeur négative) ; le symbole - provoque l'édition d'un espace si la valeur est positive et d'un caractère - si la valeur est négative.

Exemple :

```
E3 PIC +$99      +$31
E4 PIC $99      $31
E5 PIC 99-      31b
```

MOVE 31 TO E3 E4 E5 donne les résultats respectifs

- Les symboles CR et DB, exclusifs des symboles + et -, se placent en **dernière position** du format ; ils provoquent la génération des chaînes respectives "CR" et "DB" si la valeur est **négative** ; sinon ils sont remplacés par **deux** espaces.

Exemple :

```
E6 PIC 99DB

MOVE +31 TO E6 donne 31bb
MOVE -31 TO E6 donne 31DB
```

2.4. Les symboles d'insertion flottante :

Ce sont le symbole monétaire (\$) et les caractères d'édition du signe (+ et -) ; l'insertion flottante est indiquée par la présence d'au moins **deux** occurrences consécutives de l'un des trois symboles cités. Ces trois symboles s'excluent mutuellement dans ce cas.

- La chaîne des symboles "flottants" ne peut être précédée que de symboles d'insertion simple ou fixe et éventuellement du point décimal ; elle ne peut être entrecoupée que par des symboles d'insertion simple ou par le point décimal ; dans ce dernier cas, la partie fractionnaire de la valeur doit être décrite en totalité par le même symbole "flottant".
- Une occurrence d'un de ces symboles représente **un chiffre** ; les zéros qui figureraient à gauche sont remplacés par des espaces sauf celui qui précède le premier chiffre non nul : celui-ci sera remplacé par \$ ou + ou - (règles identiques à celle de l'insertion fixe).

Exemple 1 :

```

E7      PIC      $$$99      décrit une rubrique de 5 caractères

MOVE 500 TO E7  donne  b$500
MOVE 1000 TO E7  donne $1000
MOVE 1 TO E7  donne  bb$01

```

Dans le dernier cas, l'édition des chiffres est "forcée" à la rencontre du premier symbole 9, d'où le zéro.

Exemple 2 :

```

E8  PIC  $(4).$$  décrit une rubrique de 7 caractères

MOVE 50.1 TO E8  donne  b$50.10
MOVE 0.01 TO E8  donne  bbb$.01

```

Dans le dernier cas, l'édition est forcée par le point décimal et par le fait que la partie fractionnaire est non nulle ; le chiffre 0 édité a un sens et ne peut être supprimé.

```

MOVE 0 TO E8      donne 7 espaces

```

A noter que \$(4).\$9 est un format incorrect.

- Si la chaîne des symboles "flottants" contient des symboles d'insertion simple, ceux-ci sont remplacés par le caractère qui leur correspond à condition de ne pas figurer dans une partie de remplacement des zéros de gauche ; sinon ils sont systématiquement remplacés par des espaces.

Exemple 3 :

```

E9  PIC  ++,++9.99  décrit une rubrique de 9 caractères

MOVE 1000 TO E9  donne  +1,000.00
MOVE -50 TO E9  donne  bbb-50.00
MOVE 0 TO E9  donne  bbbb+0.00

```

3. Les symboles de suppression et de remplacement

Le principe est le même que celui que nous venons d'étudier pour l'insertion flottante excepté le fait que tous les zéros concernés sont remplacés par le même caractère.

3.1. Remplacement par des espaces

Il s'agit du symbole **Z** ; il représente un chiffre.

Exemple 1 :

E10	PIC Z(5)	décrit une rubrique de 5 caractères
MOVE 500	TO E10	donne bb500

Exemple 2 :

E11	PIC Z(5).ZZ	décrit une rubrique de 8 caractères
MOVE 500	TO E11	donne bb 500.00
MOVE 0.05	TO E11	donne bbbb .05
MOVE 0	TO E11	donne 8 espaces

Exemple 3 :

E12	PIC ZZ,ZZZ	décrit une rubrique de 6 caractères
MOVE 1000	TO E12	donne b 1,000
MOVE 20	TO E12	donne bbb 20
MOVE 0	TO E12	donne 6 espaces

3.2. Remplacement par des astérisques

Il s'agit du symbole ***** ; son utilisation est identique à celle du symbole Z.

Exemple 1 :

E13	PIC *(5)	décrit une rubrique de 5 caractères
MOVE 100	TO E13	donne **100
MOVE 0	TO E13	donne *****

Exemple 2 :

E14	PIC *(5).**	décrit une rubrique de 8 caractères
MOVE 0	TO E14	donne *****

4. Remarques générales

- ◆ la longueur d'une rubrique d'édition est constante et correspond au nombre de symboles du format.
- ◆ parmi les dispositifs étudiés, certains s'excluent mutuellement ainsi, pour un format donné :
 - les symboles Z et * et d'insertion s'excluent.
 - on ne peut faire qu'une seule spécification du signe (par + ou - en insertion fixe à gauche ou à droite, par CR ou DB, par + ou - en insertion flottante).
- ◆ insertion fixe et insertion flottante du symbole monétaire s'excluent.
- ◆ Le panachage à l'intérieur d'un même format des symboles d'édition et du symbole 9 est possible, à condition de respecter les deux règles suivantes :
 - 1) Les symboles d'édition doivent figurer en début de format, les symboles 9 en fin de format.
 - 2) La partie fractionnaire de la valeur éditée doit être décrite en totalité à l'aide du même symbole.

5. Rubrique numérique éditée

Une rubrique numérique éditée peut être décrite à l'aide de tous les symboles étudiés.

Il faut toutefois se souvenir que la valeur qu'elle contient est assimilée à une chaîne de caractères ; ce fait a une incidence sur l'utilisation des instructions arithmétiques, de l'instruction MOVE et de la clause VALUE

Exemple 1 :

```
E15 PIC *(4)9.99 VALUE "***21.50".
```

littéral non numérique

L'utilisation conjointe de la clause BLANK est possible à condition que le format ne contienne pas de symbole *.

6. Rubrique alphanumérique éditée

Une rubrique alphanumérique éditée ne peut être décrite que par les symboles A, X, 9 et les symboles d'insertion simple B, 0, /

Exemple : E16 PIC XXBXXBXX

```
MOVE "ABCDEF" TO E15 donne ABbCDbEF
```

7. Modification du sens de certains symboles d'édition

Les règles standards appliquées sont celles du système numérique anglo-saxon. Il est possible de réaliser deux adaptations pour se rapprocher du système français.

Elles se font dans le paragraphe `SPECIAL-NAMES` de l'`ENVIRONMENT DIVISION`.

7.1. Changement de symbole monétaire

```
CURRENCY SIGN IS lit
```

lit est un littéral non numérique d'un caractère, différent des chiffres, des caractères spéciaux, de l'espace et des lettres A, B, C, D, P, R, S, V, X, Z.

`CURRENCY IS "F"` oblige le programmeur à remplacer toutes les occurrences du caractère \$ dans les formats numériques édités par la lettre F.

Exemple : `E17 PIC F(5)9.99`

7.2. Inversion des rôles de la virgule et du point

```
DECIMAL-POINT IS COMMA
```

Dans les formats édités et les littéraux numériques, le point devient un symbole d'insertion simple et c'est la virgule qui assure l'alignement sur la marque décimale.

Exemple : `E18 PIC Z.ZZZ.ZZ9,99`

```
MOVE 1500000 TO E15 donne 1.500.000,00
```

8. Gestion de page

8.1. Définition de la page logique

Il est possible de définir une PAGE LOGIQUE en termes de nombres de lignes. Une page logique comprend 3 zones :

- une marge supérieure de page (TOP)
 - un corps de page (LINAGE)
 - une marge inférieur de page (BOTTOM)
- ◆ On appelle pied de page (FOOTING) un certain nombre de lignes défini en fin de corps. La définition de page se fera dans la phrase FD associée au fichier d'édition par les clauses:

$$\begin{array}{c} \underline{\text{LINAGE}} \text{ IS } \left\{ \begin{array}{l} \text{entier-1} \\ \text{nd-1} \end{array} \right\} \text{ LINES} \quad \left[\text{ WITH } \underline{\text{FOOTING}} \text{ AT } \left\{ \begin{array}{l} \text{entier-2} \\ \text{nd-2} \end{array} \right\} \right] \\ \left[\text{ LINES AT } \underline{\text{TOP}} \left\{ \begin{array}{l} \text{entier-3} \\ \text{nd-3} \end{array} \right\} \right] \quad \left[\text{ LINES AT } \underline{\text{BOTTOM}} \left\{ \begin{array}{l} \text{entier-4} \\ \text{nd-4} \end{array} \right\} \right] \end{array}$$

- ◆ Les entiers (ou les rubriques entières non signées) indiquent le nombre de lignes composant la zone concernée ; la valeur par défaut de TOP et BOTTOM est zéro, celle de FOOTING est égale au corps de page.
- ◆ La page logique est un concept indépendant de la page physique ; le fichier d'édition est formé d'une suite de pages logiques CONTIGUES.
- ◆ La clause LINAGE entraîne la génération automatique d'un compteur de lignes, LINAGE-COUNTER, contenant à tout moment le n° de ligne auquel l'imprimante est positionné à l'intérieur du corps de page courant. Il doit être qualifié par le nom du fichier s'il y a plusieurs fichiers d'édition il est réinitialisé à 1 à chaque changement de page logique.

8.2. Utilisation

Les locutions TOP et BOTTOM définissent les espacements (éventuellement variables) entre 2 pages consécutives. Les valeurs associées à LINAGE et FOOTING permettent de détecter respectivement le débordement de page et la fin de page et sont utilisées en relation avec la locution END-OF-PAGE d'une instruction WRITE :

$$\begin{array}{c} \underline{\text{WRITE}} \dots \quad \text{AT } \left\{ \begin{array}{l} \underline{\text{END-OF-PAGE}} \\ \underline{\text{EOP}} \end{array} \right\} \text{ phrase impérative-1} \\ \left[\text{ NOT } \left\{ \begin{array}{l} \text{AT } \underline{\text{END-OF-PAGE}} \\ \underline{\text{EOP}} \end{array} \right\} \text{ phrase impérative-2} \right] \end{array}$$

La phrase impérative est exécutée, après que soit effectuée l'écriture :

- ◆ dès que l'on atteint la zone FOOTING (i.e. dès que la valeur de LINAGE-COUNTER est supérieure ou égale à entier-2)
- ◆ ou si l'instruction fait déborder de la page (i.e. dès que LINAGE-COUNTER est supérieur à entier-1).

Exemple :

```
FD IMPRIM LABEL RECORD STANDARD
      LINAGE 50 TOP 6 BOTTOM 10.
01 LIGNE.
   -
   -
   -
   -
      WRITE LIGNE AT EOP PERFORM FIN-PAGE.
   -
   -
FIN-PAGE. PERFORM SOUS-TOTAL ...
      WRITE LIGNE FROM EN-TETE AFTER PAGE.
```

N-B : 1) On pourrait aussi tester après l'écriture d'une ligne "normale" :

```
WRITE LIGNE
      IF LINAGE-COUNTER > 44 PERFORM FIN-PAGE.
```

2) COBOL prévoir un module d'édition (REPORT-WRITER) complet permettant la description de l'aspect physique et la production d'états structurés.

Exercices

	EMETTEUR		RECEPTEUR	
	CONTENU	IMAGE	IMAGE	CONTENU
A	12345	S9 (5)	-ZBZZ9V, 99	
B	00345	S9 (5)	\$ZBZZ9V, 99	
C	00000	S9 (5)	-ZBZZ9V, 99	
D	00000	S9 (5)	\$ZBZZ9V, 99	
E	00000	S9 (5)	ZBZZZ	
F	12345	9 (3) V99	\$ZBZZ9V, 99	
G	12345	9 (5)	\$***9V, 99	
H	00234	9 (5)	\$**B**9V, 99	
I	00000	9 (5)	\$*****V, 99	
J	00000	9 (5)	\$***9V, 99	
K	12345	99V9 (3)	\$**B**9V, 99	
L	12345	9 (5)	\$\$\$B\$\$9V, 99	
M	00345	9 (5)	\$\$\$, \$\$9.99	
N	00000	9 (5)	\$\$\$B\$\$9V, 99	
O	12345	9 (4) V99	\$\$\$B\$\$9V, 99	
P	12345	V9 (5)	\$\$\$B\$\$9V, 99	
Q	-12345	S9 (5)	-ZBZZ9V, 99	
R	-00345	S9 (5)	-ZBZZ9V, 99	
S	00123	S9 (5)	---B---V, 99	
T	-00005	S9 (5)	---B---V, 99	
U	12345	S9 (5)	+ZZZZZV, 99	
V	-12345	S9 (5)	+ZBZZZV, 99	
W	00345	S9 (5)	+++++V, 99	
X	12345	9 (5)	BB999V, 99	
Y	12345	S9 (5)	ZBZZZV, 99CR	
Z	00234	S9 (5)	ZBZZZV, 99CR	